
What Is The
Image Base Option?

QShould I bother specifying an
image base for my Delphi 2

applications and DLLs?

AThe Project | Options |
Linker | Image base: option

specifies where in memory
Windows will attempt to load the
DLL or EXE: it defaults to $00400000.
Bearing in mind that each Windows
process is individually mapped
into the same address range, a pro-
gram will always be safe to load at
this address. However once the
program has loaded there, any sub-
sequent DLLs cannot, so the image
base should be changed. If the de-
fault value is used, the DLL takes a
bit longer to load as Windows must
move it to a different address. The
advised range for DLLs is $40000000
to $7FFFFFFF, but the lower four dig-
its are ignored and should be zero.

Each DLL should be given a
unique image base address in the
context of the application, but you
should also take into account other
DLLs that will be used, for example
the BDE. If you look at all the DLLs
supplied with the 32-bit BDE with
Windows 95 Quick View, or with
TDump, their Image Base values
are all different. Borland strive to
give unique values for every DLL
they manufacture to stop possible
performance penalties. Interest-
ingly, the image base for the main
BDE DLL, IDAPI32.DLL is $4BDE0000.

Floating Point Inaccuracy

QI use an InterBase table with
a FLOAT field. When I display

the value of this field in a data-
aware control the value is more
precise than what was stored, but
doesn’t correspond to the value

which should be displayed. For
example, a value stored as 0.18 is
displayed in my Delphi program as
0.180000007152557.

This seems to be a standard
rounding issue. We use digital com-
puters which operate in binary. We
humans tend to work in decimal (to
do with fingers and thumbs I
gather). Given the number of bits a
floating point type has (there are
several types and their storage
sizes vary), certain values can’t be
represented exactly in base 2. So
the bits are set to store a value as
close as possible to the value de-
sired. That’s why the field type
FLOAT is described as having only
up to seven digits of precision – the
number of bits used can only accu-
rately represent numbers with that
number of digits in binary.

You can tell the database compo-
nents to only display a certain
precision using the Precision prop-
erty, but if you programmatically
compare floating point numbers,
you tend to need to use an error
factor. Something like:

const
  ErrorFactor = 0.000001;
  { varies depending on
    floating point type }

Then, instead of

if FloatVar = 0.18 then...

you use

if Abs(FloatVar - 0.18) <
  ErrorFactor then...

InterBase BLOb
SubTypes And Segments

QWhat are InterBase BLOb
subtypes and segments and

how are they used?

ARather than launching into a
discussion of these myself, I

refer you to the InterBase Program-
mer’s Guide. You get this with most
versions of InterBase (except the
local InterBase that comes with
Delphi 1). Delphi 2 users can find
an online version of this manual by
bringing up the Local InterBase
Help help file. From the Contents
page, choose Programmer’s Guide.
For information on the topic in
question, choose Working with
BLOb Data and then either BLOb
subtypes or BLOb Segment Length.

Raw Printer Access

QI used to use the Win16
Escape routine with a

PassThrough or DeviceData parame-
ter to send raw data to the printer.
The Delphi 2 online help says these
parameters are obsolete in Win32.
Additionally, the help and declara-
tion for the SpoolFile API, used to
send a binary printer file directly to
a printer, are gone in Delphi 2.
What do we use instead?

AThe Win32 API provides sup-
port for directly writing to a

printer which removes problems
with these two older approaches.
Listing 1 contains a translation of a
routine from the MSDN (article
Q138594) that sends raw data to a
printer. The Win16 problem areas
are highlighted and expanded
upon in further MSDN articles:
Q111010, Q35708, Q139011.

This routine can be called with
any variable that holds some
appropriate data. As a simple
example, to send two form feeds
(where a form feed is ASCII charac-
ter 12) to an HP DeskJet printer,
you can use this (remember, the
names of all installed printers
can be obtained from the

Printer.Printers property, a
TStrings object):

const
  Data: array[1..2] of Char =
    #12#12;
RawDataToPrinter(
  ’HP DeskJet 520 Printer’,
  Data, SizeOf(Data));

if you wanted to write to the default
printer on LPT1: you could use:

RawDataToPrinter(’LPT1’,
  Data, SizeOf(Data));

Virtual And
Override Clarification

QWhy does Delphi have two
keywords for polymorphism

(virtual and override), where
previous Borland Pascal products
and C++ have just one (virtual)?

ADelphi actually has three
keywords (virtual, dynamic

and override) but let’s not compli-
cate matters just yet. Before look-
ing at the question itself, let’s have
a brief recap on what polymor-
phism is and how it works. Poly-
morphism allows a situation where
you can call a method of an object,
but at run-time the object may be
an instance of any of several differ-
ent classes, all from one branch of
your object hierarchy. Because the
compiler can’t know which class
will be present at run-time, it can’t
calculate the address to branch
execution off to. Instead, if told to,
it needs to call code to perform
some lookup at run-time, and get
the appropriate class’s code to
run.

In C++ and Delphi the way we can
introduce a new polymorphic rou-
tine into a base class is by adding
the word virtual onto its declara-
tion. If we make a new class inher-
ited from the base class and wish
to define new functionality for this
virtual method, C++ lets us re-
declare the method and again use
the virtual keyword. In Delphi we
must use the override keyword.

There are two distinct opera-
tions here: adding a new polymor-
phic routine and re-declaring an
existing polymorphic routine. C++

does both operations with one
keyword, Delphi distinguishes
between the two with different key-
words. Look at Listing 2, from the
CARS.DPR project, for a Delphi
example of a virtual and over-
ridden routine. This allows code
written like this to work:

var Cars: array[1..2] of TCar;
...
Cars[1] := TCar.Create;
Cars[2] := TRacingCar.Create;
Cars[1].Drive;
Cars[2].Drive;
Cars[1].Free;
Cars[2].Free;

Even though the array is defined to
be of two TCar objects, it is quite
acceptable to place objects de-
rived from TCar in it. Any TCar or
TCar-descendent could be placed in
the array at any time during the
execution of the program. The
compiler has no chance of predict-
ing which object will be there and
so has to generate code which will
find that out at run-time and act
accordingly.

The two calls to the Drive
method work because the virtual
and override keywords have been
used to enable polymorphism to
work, but how do they function?

This business of waiting until
run-time to find out what to exe-
cute is called late binding, where
the normal compiler job of decid-
ing what needs to execute at com-
pile time is called early binding.
Late binding is implemented in
Delphi and C++ by the use of lookup
tables of addresses called virtual
method tables (VMTs). Each class
has a VMT made by the compiler
which is stored on the heap (in the
old object model used by Borland
Pascal it was stored in the data
segment, rather restricting the to-
tal number of classes). The VMT
includes the addresses of all
virtual methods for that class and,
when you call a virtual method,
the generated code effectively
does an indirect jump by jumping
to the address stored at the
relevant offset in the VMT.

Using the word virtual in Delphi
adds a new entry onto the VMT,
whereas override changes an entry

in the VMT. This differs from C++
and the old Pascal object model
where virtual will add a new VMT
entry if there is not already one for
a method of that name, otherwise
virtual changes the relevant VMT
entry. Pictorially, the TCar and
TRacingCar VMT are shown in
Figure 1.

Notice that each class’s VMT has
all virtual methods for that class
and its ancestors. It’s worth bear-
ing in mind that had we tried to use
virtual in both the ancestor and
descendent (as we would in C++)
we would not get the desired re-
sult. Instead, the VMT would look
as in Figure 2. The use of virtual
twice has added two new entries,
despite the two methods having
the same name.

I said at the start that there was
a third keyword: dynamic. This can
be used instead of virtual and
adds an entry to a dynamic method
table (DMT). A DMT is much the
same as a VMT but for one differ-
ence. Where a VMT stores entries
for all virtual methods including
those inherited from ancestors, a
DMT only includes entries for
dynamic methods defined in that
class. The implication of this is that
DMTs are much smaller than
VMTs. Delphi message handlers
are implemented as dynamic meth-
ods. There could be many message
handlers in all the ancestors of any
given class, but only that class’s
own message handlers are stored
in the DMT to stop it swelling
excessively. However, there is an-
other implication for DMTs. In or-
der to work, the code for dynamic
methods might need to search the
DMT for the given class, and also
for many of its ancestors, before
finding the relevant entry. It is
therefore potentially much slower.

For methods where there is no
efficiency requirement, you can
make methods dynamic instead of
virtual. The override keyword is
used when redefining both virtual
and dynamic methods.

So, back to the original question.
Why the change from the estab-
lished way of doing things? It was
realised that adding the override
keyword would provide greater
version resiliency in classes. Let’s

consider a few possible scenarios.
Firstly, a user of Delphi 1’s VCL
creates a descendant of TEdit and
adds a new method, Foo, as virtual.
In Delphi 2’s VCL, Borland adds a
virtual method to TEdit, called Foo.
Were it not for the override syntax,
the user’s code would break when
they changed version, since they
would now be redefining a VCL
method.

The second, but rather less
likely, scenario involves Delphi 1’s
VCL having a method called Bar
that is virtual and a user overrides
this method in a descendant class.
In Delphi 2’s VCL, Borland decides
to make that method dynamic since
it isn’t a performance critical
section. As in the first scenario, the
user’s code would break if the
override syntax weren’t there,
since they would have to use
virtual for a re-declared virtual
method and dynamic for a
re-declared dynamic method.

The first scenario is principally
the reason override was added to

TCar = class
public
  {Declare new polymorphic routine}
  procedure Drive; virtual;
end;
TRacingCar = class(TCar)
public
  { Redefine existing polymorphic
    routine }
  procedure Drive; override;
end;
procedure TCar.Drive;
begin
  ShowMessage(’Chug, chug, chug’);
end;
procedure TRacingCar.Drive;
begin
  ShowMessage(’Vroomm, vroommm’);
end;

➤ Listing 2

// RawDataToPrinter - sends binary data directly to a printer
// Params: PrinterName - printer name
//         Data        - raw data bytes, any variable will do
//         Count       - length of Data in bytes
// Gives an exception upon failure
// (with Break on Exception on it may give several)
procedure RawDataToPrinter(PrinterName: String;
   const Data; Count: Integer);
implementation
uses WinSpool, Windows, Consts, SysUtils;
type EPrinterError = class(Exception);
procedure Error;
begin
  raise EPrinterError.CreateRes(SInvalidPrinterOp);
end;
procedure RawDataToPrinter(PrinterName: String;
 const Data; Count: Integer);
type
  TDoc_Info_1 = record
    DocName,
    OutputFile,
    Datatype: PChar;
  end;
var
  hPrinter: THandle;
  DocInfo: TDoc_Info_1;
  BytesWritten: Integer;
begin
  // Need a handle to the printer

  if not OpenPrinter(PChar(PrinterName), hPrinter, nil) then
    Error;
  // Fill in the structure with info about this “document”
  DocInfo.DocName := ’Document’;
  DocInfo.OutputFile := nil;
  DocInfo.Datatype := ’RAW’;
  try
    // Inform the spooler the document is beginning
    if StartDocPrinter(hPrinter, 1, @DocInfo) = 0 then Error;
    try
      // Start a page
      if not StartPagePrinter(hPrinter) then Error;
      try
        // Send the data to the printer
        if not WritePrinter(hPrinter, @Data, Count,
          BytesWritten) then Error;
      finally
        // End the page
        if not EndPagePrinter(hPrinter) then Error;
      end;
    finally
      // Inform the spooler that the document is ending
      if not EndDocPrinter(hPrinter) then Error;
    end;
  finally
    // Tidy up the printer handle
    ClosePrinter(hPrinter);
  end;
  // Check to see if correct number of bytes written
  if BytesWritten <> Count then Error;
end;

➤ Listing 1 

TObject.FreeInstance

TCar.Drive

TCar’s VMT

➤ Figure 1

➤ Figure 2

the language. Introducing new
virtual methods in base classes
happens fairly regularly between
major product versions. The
second scenario is almost inciden-
tal as methods are rarely changed
from virtual to dynamic. The im-
portant point is that the use of
override takes away the problem of
brittle classes, and makes your
classes more tolerant of base class
changes. This means your compo-
nent classes are less likely to re-
quire modification to work in
subsequent versions of Delphi.
This also gives Borland, and other
authors of widely used base
classes, more scope to enhance the
base classes over time.

In short, virtual or dynamic and
override provide a superior
solution to just virtual or dynamic
alone.

Note that the term resiliency is
being used here to mean source
code resiliency across versions,
not DCU binaries. If a change is
made to the interface of a base
class, all descendent source code
must be re-compiled. However,

because of the existence of
override, the descendent source
should not necessarily need
modifying first.

